Ecuaciones Diferenciales

Maxima tiene dos comandos para resolver EDs:

- ode2 (ver pestaña "Ecuaciones -> Resolver EDO") resuelve ecuaciones de manera exacta (cuando sabe encontrar la solución)
- plotdf resuelve ecuaciones con un método numérico, dibujando la gráfica de la solución

Ejemplo 1: Resolver x'(t) = x(t)^2, con x(0)=10

Conviene definir la ecuación que queremos resolver

```
--> ed1:'diff(x,t)=x^2;
```

(%o1) $x\frac{d}{d\,t}=x^2$

MÉTODO 1: usar la pestaña "Ecuaciones -> Resolver EDO"

```
--> ode2(ed1, x, t);
```

(%o2) $-\frac{1}{x}=t+\%c$

Para determinar la "c" usar la pestaña "Ecuaciones -> Problema valor inicial (1)"

```
--> ic1(%, t=0, x=10);
```

(%o3) $-\frac{1}{x}=\frac{10\,t-1}{10}$

Para despejar "x" usar solve

```
--> solve([%], [x]);
```

(%o4) $[x=-\frac{10}{10\,t-1}]$

La solución queda

--> x(t):= 10/(1-10*t);

(%o5) $x(t):=\frac{10}{1-10\,t}$

--> wxplot2d([x(t)], [t,0,.09],[y,0,100])$

(%t6) 

MÉTODO 2: Si el método anterior no funciona (Maxima no sabe encontrar una solución exacta, o no sabe despejar x), se puede usar el comando "plotdf".

"plotdf" no tiene pestaña, y hay que escribir el comando con sus correspondientes opciones.
Por ejemplo, para la ED

x'(t)=x(t)*(1-x(t)) se escribe plotdf(x*(1-x),[t,x],[t,0,5],[x,0,2])

Se abre una pantalla de tipo plot, y pinchando en un punto se dibuja la gráfica de la solución que pasa por ese punto

--> plotdf(x*(1-x),[t,x],[t,0,5],[x,0,2]);

(%o7) 0

Figura 1: C:\verso\curso_UM\bioquim2013\ejercicios\fig_edlog.jpg



si queremos sólo la solución que pasa por el punto x(0)=1.2, podemos usar la opción [trajectory_at, 0, 1.2]

(ver más opciones en la pestaña en la pestaña Ayuda -> Indice -> plotdf)

--> plotdf(x^2,[t,x],[t,0,0.1],[x,0,100], [trajectory_at, 0, 1.2]);

(%o8) 0

Ejemplo 2: Ecuación logística

x'(t) = 0.2*x(t)*[5-x(t)] con x(0)= 1, y con x(0)= 6

MÉTODO 1: Probamos primero la pestaña "Ecuaciones --> Resolver EDO"

--> ed2: 'diff(x,t)=0.2*x*(5-x);ratprint:false;

(%o1) $\frac{d}{d\,t}x=0.2\,(5-x)\,x$

(%o2) *false*

--> ode2(ed2, x, t);

(%o3) $\log(x)-\log(x-5)=t+\%c$

--> ic1(%, t=0, x=1);

(%o4) $\log(x)-\log(x-5)=t-\log(-4)$

--> logcontract(%);

(%o5) $\log\left(\frac{x}{x-5}\right)=t-\log(-4)$

--> solve([%], [x]);

(%o6) $[x=\frac{5\,\%e^{t}}{\%e^{t}+4}]$

--> x1(t):=(5*%e^t)/(%e^t+4);

(%o7) $x1(t):=\frac{5\,\%e^{t}}{\%e^{t}+4}$

NOTAS:

(i) Hemos usado la pestaña Simplificar --> Contraer Logaritmos antes de despejar x
(de otro modo Maxima suele dar error al despejar la x...)

(ii) Maxima puede manejar expresiones como log(-4) y luego despejar correctamente la x.
Si esto diera problemas, o aparecieran soluciones complejas, se puede usar previamente el comando "assume(0 < x and x < 5)"

se procede similarmente en el caso x(0)=6

```
--> ic1(log(x)-log(x-5)=t+%c, t=0, x=6);
```

(%o8) $\log(x)-\log(x-5)=t+\log(6)$

```
--> logcontract(%);
```

(%o9) $\log\left(\frac{x}{x-5}\right)=t+\log(6)$

```
--> solve([%], [x]);
```

(%o10) $[x=\frac{30\ \%e^{t}}{6\ \%e^{t}-1}]$

```
--> x2(t):=(30*%e^t)/(6*%e^t-1);
```

(%o11) $x2(t):=\frac{30\ \%e^{t}}{6\ \%e^{t}-1}$

```
--> wxplot2d([x1(t),x2(t)], [t,0,5],[y,0,7])$
```

(%t12) 

MÉTODO 2: usamos el comando plotdf, pinchando en los puntos x(0)=1, x(0)=6

```
--> plotdf(0.2*x*(5-x),[t,x],[t,0,5],[x,0,7]);
```

(%o1) 0

Figura 2: C:\verso\curso_UM\bioquim2013\ejercicios\fig_edlog2.jpg



Observando la gráfica y usando el cursor se puede responder a distintas preguntas

(i) a largo plazo las poblaciones tienden a 5

(ii) tras aprox 1'3 dias tiene un pto inflexión
(iii) tras aprox 3'5 días la población alcanza el 90%, etc...

□jemplo 3: ecuación logística con caza

x'(t) = 0.2*x(t)*[1-x(t)/100] - a

Resolver para x(0)=20, donde "a" es un parámetro entre 0 y 4. Dibujar la solución para distintos valores del parámetro, y determinar para qué valores de "a" se extingue la población.

□l método recomendado es el comando plotdf con la orden "sliders", que permite ir moviendo los valores del parámetro, y visualizar cómo varía la gráfica

```
--> plotdf(0.2*x*(1-x/100)-a,[t,x],[t,0,100],[x,0,100],[trajectory_at,0,20], [parameters, "a=0"], [sliders, "a=0:4"]);
```

```
(%o16) 0
```

se aprecia que:

- si a < 3'2 la población sigue creciendo hasta alcanzar a largo plazo aprox 80 indiv
- si a = 3'2 la población permanece estacionaria
- si a > 3'2 la población se extingue tras pocos meses

ver gráfica abajo con a=3'2, y varios valores de x(0) entre 15 y 25

Figura 3: C:\verso\curso_UM\bioquim2013\ejercicios\fig_ed3.jpg



Created with wxMaxima.